# $Id: PKGBUILD 84283 2013-02-15 14:33:33Z lfleischer $
# Maintainer: Lukas Fleischer <archlinux at cryptocrack dot de>
# Contributor: Aaron Bull Schaefer <aaron@elasticdog.com>
# Contributor: Ron Huang <ronhuang+aur at gmail dot com>

pkgname=fcgiwrap
pkgver=1.1.0
pkgrel=1
pkgdesc='A simple server for running CGI applications over FastCGI.'
arch=('i686' 'x86_64')
url='http://nginx.localdomain.pl/wiki/FcgiWrap'
license=('MIT')
depends=('fcgi' 'systemd')
source=("https://github.com/gnosek/fcgiwrap/archive/${pkgver}.tar.gz"
        'LICENSE')
md5sums=('d14f56bda6758a6e02aa7b3fb125cbce'
         '5aee62c27b4308f25ab32f05da387366')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  autoreconf --install
  ./configure --prefix=/usr --mandir=/share/man
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  make DESTDIR="${pkgdir}" install
  install -Dm0644 ../LICENSE "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
